Title

Finite Element Analysis for Solid Rocket Motor Cases

Authors

Herrmann, Leonard
Taylor, Robert
Green, David

Publication Date

1967-03-01

REPORT NO. 67-4

STRUCTURES AND MATERIALS RESEARCH
DEPARTMENT OF CIVIL ENGINEERING

FINITE ELEMENT ANALYSIS FOR SOLID ROCKET MOTOR CASES

by
L. R. HERRMANN
R. L. TAYLOR
and
D. R. GREEN

Interim Technical Report
Stanford Research Institute
Menlo Park, California
Subcontract No. B-87010-US

MARCH, 1967

STRUCTURAL ENGINEERING LABORATORY
UNIVERSITY OF CALIFORNIA
BERKELEY CALIFORNIA

Structures and Materials Research
Department of Civil Engineering
Division of Structural Engineering
and Structural Mechanics

Report Number 67-4

FINITE ELEMENT ANALYSIS FOR SOLID ROCKET MOTOR CASES

by

L.R. Herrmann
Assistant Professor of Civil Engineering
University of California at Davis

R.L. Taylor
Assistant Professor of Civil Engineering
University of California at Berkeley

and

D.R. Green
Lecturer in Civil Engineering
University of Glascow, Scotland

Interim Technical Report
Stanford Research Institute
Menlo Park, California

Subcontract No. B-87010-US
Under Prime Contract No. N123(60530)-55641A
with U.S. Naval Ordinance Test Station
China Lake, California

Structural Engineering Laboratory
University of California
Berkeley, California

March 1967

CONTENTS

		Page
	Preface	i
	Nomenclature	ii
1)	Introduction	1
2)	Shells of Revolution	3
3)	Cylindrical Shells (Plane Strain)	7
4)	Variational Theorem for Shell Analysis	9
5)	Finite Element Solution for Shells of Revolution	11
6)	Finite Element Solution for Cylindrical Shells	16
7)	Numerical Example	18
8)	Conclusions	19
	References	20
	Figures	21
	Table I	25
	Appendix A Computer Program Listing	A-1
	Appendix B Computer User's Manual	B-1

PREFACE

Support for the work contained herein has been provided in part by Mathematical Sciences Northwest, Seattle, Washington and by Stanford Research Institute, under Subcontract No. B-87010-US under prime contract number N123(60530)-55641A for Naval Ordnance Test Station, China Lake, California.

Nomenclature

r, z	-	Cylindrical coordinates
R	-	Radius of Curvature of Shell Meridian
s	-	Shell Coordinate along Meridian
u, w	-	Displacements of Shell
q_i	-	Loading Components
N_{ij}, M_{ij}, Q_i	-	Shell Force Resultants
ϵ_{ij}	-	Middle Surface Strains
χ_{ij}	-	Middle Surface Change in Curvatures
A_{ij}, D_{ij}	-	Shell Material Properties
C_{ij}	-	Elastic Material Moduli
α_{ij}	-	Coefficients of Linear Expansion
T	-	Temperature
V	-	Functional for Variational Theorem

1. INTRODUCTION

During the past few years considerable attention has been devoted to analysis of solid rocket grain configurations. One of the methods commonly employed is the finite element analysis of an idealized grain-case configuration. The procedure usually consists of dividing the region of interest into polygonal "elements" in the continuum (grain) and straight line "elements" in the shell (motor case). Then appropriate displacement approximations are utilized in each element and a Ritz solution is obtained for the entire configuration. To date most analyses utilize displacement fields in the continuum which keep the boundaries of the elements straight during deformation. Then, by matching the nodes of intersection between elements, a continuous displacement field is maintained. It is known that convergence of the method occurs with decrease in element size when displacements maintain compatibility between elements.

In the analysis of the shell, most of the reported literature is based on a displacement formulation, wherein the transverse and plane displacements are expanded in polynomials [1], [2]. Popov uses exact displacements for conoids which are loaded at their ends only [3]. When utilized as a motor case for a grain which is analyzed by a finite element method, all of these displacement formulations are incompatible with grain displacement fields. In the present work, a development is obtained which has a compatible displacement field with the continuum elements commonly employed. This development also uses straight segments as the basic element and, consequently, neglects the effects of initial curvature along the generator defining the elements. The errors involved in using straight segments should not be serious provided the element length to initial radius of curvature ratio is small.

With the present development, it should now be possible to investigate the effects of grain-case incompatibility. Furthermore, with this development it is not necessary to deduce "special" continuum elements at the interface as would be necessary with previous developments; hence it is easily adapted to existing computer analysis routines for motor grains.

2. SHELLS OF REVOLUTION

The equations which govern the behavior of a shell of revolution subjected to axisymmetric deformations are expressed by: (1) the three equilibrium equations [4]

$$\frac{1}{r}\left[\frac{\partial}{\partial s}(rN_{11}) - \cos\varphi\, N_{22} + \frac{r}{R} Q_1 + rq_1\right] = 0$$

$$\frac{1}{r}\left[\frac{\partial}{\partial s}(rQ_1) - \sin\varphi\, N_{22} - \frac{r}{R} N_{11} + rq_n\right] = 0 \qquad (2.1)$$

$$\frac{1}{r}\left[\frac{\partial}{\partial s}(rM_{11}) - \cos\varphi\, M_{22} - rQ_1\right] = 0$$

(2) the reference surface strain-displacement-temperature measures

$$\epsilon_{11} = \frac{\partial u}{\partial s} + \frac{w}{R} - \alpha_{11}\,\Delta T$$

$$\epsilon_{22} = \frac{1}{r}[u\cos\varphi + w\sin\varphi] - \alpha_{22}\,\Delta T \qquad (2.2)$$

$$\chi_{11} = \frac{\partial}{\partial s}\left(\frac{\partial w}{\partial s} - \frac{u}{R}\right)$$

$$\chi_{22} = \frac{\cos\varphi}{r}\left(\frac{\partial w}{\partial s} - \frac{u}{R}\right)$$

where

$$\Delta T = T - T_o$$

T is the solution to the uncoupled Fourier's heat conduction equation; and (3) the constitutive equations wherein the stress resultants are related to the strain-displacement-temperature measures.

$$N_{11} = A_{11}\,\epsilon_{11} + A_{12}\,\epsilon_{12} + D_{11}\left(\frac{1}{R} - \frac{\sin\varphi}{r}\right)\left(\chi_{11} + \frac{\epsilon_{11}}{R}\right)$$

$$N_{22} = A_{12}\,\epsilon_{11} + A_{22}\,\epsilon_{22} - D_{22}\left(\frac{1}{R} - \frac{\sin\varphi}{r}\right)\left(\chi_{22} + \frac{\sin\varphi}{r}\right)\epsilon_{22} \qquad (2.3)$$

and

$$M_{11} = - D_{11} K_{11} - D_{12} K_{22} - D_{11} \left(\frac{1}{R} - \frac{\sin \varphi}{r}\right) \epsilon_{11}$$

$$M_{22} = - D_{12} K_{11} - D_{22} K_{22} + D_{22} \left(\frac{1}{R} - \frac{\sin \varphi}{r}\right) \epsilon_{22}$$

where

$$A_{ij} = \int_{-h/2}^{h/2} C_{ij} \, dz$$

$$D_{ij} = \int_{-h/2}^{h/2} C_{ij} z^2 \, dz \qquad (2.4)$$

and C_{ij} are the elastic-moduli for the orthotropic material.

Equations (2.1) to (2.3) constitute the governing equations for the analysis of shells of revolution subjected to axisymmetric deformations. In order to construct a finite element solution which satisfies all the continuity requirements at element interfaces and uses only displacements as primary dependent variables, it is necessary to expand the inplane displacement u with at least two degrees of freedom and the transverse displacement w with at least four degrees of freedom within each element. Grafton-Strome [1] use a polynomial representation for u and w. If this shell element is used in connection with an elastic filler which is expressed by conventional finite elements with linear displacement expansions, incompatibilities exist between the shell and filler deformation fields. In order to retain the linear displacement expansion for the continuum element the requirement of compatibility requires linear expansion of both shell displacement components. Consequently, to allow a linear displacement expansion it is necessary to retain a third primary dependent variable. In the development to follow the meridian moment is retained as the third dependent variable.

The governing equations, in a form suitable for an analysis based on the above arguments are recast in the form:

$$\frac{1}{r}\left\{\frac{\partial}{\partial s}\left(r\left[\bar{A}_{11}\epsilon_{11}+A_{12}\epsilon_{22}-c\left(M_{11}+D_{12}\chi_{22}\right)\right]\right)-\cos\varphi\left[A_{12}\epsilon_{11}+\bar{A}_{22}\epsilon_{22}-cD_{22}\chi_{22}\right]\right.$$

$$\left.+\frac{1}{R}\left(\frac{\partial}{\partial s}(rM_{11})-\cos\varphi\left[\frac{D_{12}}{D_{11}}M_{11}-\bar{D}_{22}\chi_{22}+c\left(D_{12}\epsilon_{11}+D_{22}\epsilon_{22}\right)\right]\right)+rq_{1}\right\}=0 \quad (2.5)$$

$$\frac{1}{r}\left\{\frac{\partial}{\partial s}\left(\frac{\partial}{\partial s}(rM_{11})-\cos\varphi\left[\frac{D_{12}}{D_{11}}M_{11}-\bar{D}_{22}\chi_{22}+c\left(D_{12}\epsilon_{11}+D_{22}\epsilon_{22}\right)\right]\right)\right.$$

$$-\sin\varphi\left[A_{12}\epsilon_{11}+A_{22}\epsilon_{22}-cD_{22}\chi_{22}\right]-\frac{r}{R}\left[\bar{A}_{11}\epsilon_{11}+A_{12}\epsilon_{22}\right.$$

$$\left.\left.-c\left(M_{11}+D_{12}\chi_{22}\right)\right]+rq_{n}\right\}=0 \quad (2.6)$$

and

$$\frac{\partial\theta}{\partial s}+\frac{M_{11}}{D_{11}}+\frac{D_{12}}{D_{11}}\chi_{22}+\left(\frac{1}{R}-\frac{\sin\varphi}{r}\right)\epsilon_{11}=0 \quad (2.7)$$

where

$$c=\left(\frac{1}{R}-\frac{\sin}{r}\varphi\right)$$

$$\bar{A}_{11}=A_{11}+\frac{\sin\varphi}{r}cD_{11}$$

$$\bar{A}_{22}=A_{22}-\frac{\sin\varphi}{r}cD_{22}$$

$$\bar{D}_{22}=D_{22}-\frac{D_{12}^{2}}{D_{11}} \quad (2.8)$$

$$\chi_{11}=\frac{\partial\theta}{\partial s}$$

Equations (2.5) and (2.6) are the first two of Equation (2.1) recast in terms of the selected primary dependent variables; while Equation (2.7) is the constitutive equation for M_{11}. The interrelationships of the quantities, which occur on the boundary of the shell, in terms of the selected primary dependent variables are

$$\theta = \frac{\partial w}{\partial s} - \frac{u}{R} \tag{2.9}$$

$$Q_1 = \frac{\partial}{\partial s}(rM_{11}) - \cos\varphi\left[\frac{D_{12}}{D_{11}} M_{11} - \bar{D}_{22}\chi_{22} + c\left(D_{12}\epsilon_{11} + D_{22}\epsilon_{22}\right)\right] \tag{2.10}$$

and

$$N_{11} = \bar{A}_{11}\epsilon_{11} + A_{12}\epsilon_{22} - c\left(M_{11} + D_{12}\chi_{22}\right) \tag{2.11}$$

where θ is the change in slope of the shell along a meridian.

3. CYLINDRICAL SHELL (PLANE STRAIN)

The governing equations for a cylindrical shell in a state of plane strain along the generator are given by the three equilibrium equations

$$\frac{\partial N_{11}}{\partial s} + c\,Q_1 + q_1 = 0$$

$$\frac{\partial Q_1}{\partial s} - c\,N_{11} + q_n = 0 \qquad (3.1)$$

$$\frac{\partial M_{11}}{\partial s} - Q_1 = 0$$

the strain displacement-temperature measures

$$\epsilon_{11} = \frac{\partial u}{\partial s} + cw - \alpha_{11} \cdot \Delta T,$$

$$\epsilon_{22} = -\alpha_{22} \cdot \Delta T$$

$$\chi_{11} = \frac{\partial}{\partial s}\left(\frac{\partial w}{\partial s} - cu\right); \qquad (3.2)$$

$$\chi_{22} = 0$$

and the constitutive equations

$$N_{11} = A_{11}\epsilon_{11} + A_{12}\epsilon_{22} + c\,D_{11}(\chi_{11} + c\,\epsilon_{11})$$

$$N_{22} = A_{12}\epsilon_{11} + A_{22}\epsilon_{22} \qquad (3.3)$$

$$M_{11} = -D_{11}(\chi_{11} + c\,\epsilon_{11})$$

$$M_{22} = -D_{12}\,\chi_{11} + c\,D_{22}\,\epsilon_{22}$$

where

$$c = \frac{1}{a}, \quad a = \text{shell radius.} \tag{3.4}$$

In the same manner as for shells of revolution the equations are recast in the form

$$\frac{\partial}{\partial s}\left(A_{11}\epsilon_{11}+A_{12}\epsilon_0 - c\ M_{11}\right) + c\ \frac{\partial M_{11}}{\partial s} + q_1 = 0 \tag{3.5}$$

$$\frac{\partial^2 M_{11}}{\partial s^2} - c\left(\bar{A}_{11}\epsilon_{11}+A_{12}\epsilon_0 - c\ M_{11}\right) + q_n = 0 \tag{3.6}$$

and

$$\frac{\partial \theta}{\partial s} + \frac{M_{11}}{D_{11}} + c\ \epsilon_{11} = 0 \tag{3.7}$$

where

$$\chi_{11} = \frac{\partial \theta}{\partial s} \quad \text{and} \quad \bar{A}_{11} = A_{11} \tag{3.8}$$

The auxiliary boundary variables in terms of the dependent variables are expressed by

$$\theta = \frac{\partial w}{\partial s} - cu \tag{3.9}$$

$$Q_1 = \frac{\partial M_{11}}{\partial s} \tag{3.10}$$

$$N_{11} = \bar{A}_{11}\epsilon_{11}+A_{12}\epsilon_0 - c\ M_{11} \tag{3.11}$$

where θ is the change in the slope of the shell along s.

4. VARIATIONAL THEOREM FOR SHELL ANALYSIS

In order to develop a finite element shell analysis based upon the above formulation, it is expedient to express the shell problem by an equivalent variational equation. The shell problems defined by sections (2) and (3) with their auxiliary relations are expressed by the variational equation

$$\delta V = 0 \tag{4.1}$$

where V is the functional whose variational yields as Euler equations; equations (5) through (7), and as natural boundary conditions equations (9) through (11) of the previous two sections. The functional V is determined to have the form*

$$V = \iint_B \left\{ \frac{1}{2} \left(\bar{A}_{11}\epsilon_{11}^{\ 2} + \bar{A}_{22}\epsilon_{22}^{\ 2} + \bar{D}_{22}\chi_{22}^{\ 2} - \frac{1}{D_{11}} M_{11}^{\ 2} \right) + A_{12}\epsilon_{11}\epsilon_{22} + \frac{\partial M_{11}}{\partial s}\,\theta + \left(1 - \frac{D_{12}}{D_{11}}\right) M_{11}\chi_{22} - c\left[\left(M_{11} + D_{12}\chi_{22} \right) \epsilon_{11} + D_{22}\chi_{22}\epsilon_{22} \right] - q_n\, w - q_1\, u \right\} dv - \int_C \left(N_{11}^{\ a}\, u + Q_1^{\ a}\, w + \theta^a\, M_{11} \right) dc \tag{4.2}$$

The surface integral is to be evaluated over the entire shell B and the line integral over the boundaries where N_{11}, Q, or θ are specified.

*A Galerkin approach is used to deduce the variational theorem. The divergence theorem is then used to obtain the quadratic form given by Equation (4.2).

The symbols N_{11}^{a}, Q_{1}^{a} and θ^{a} denote specified values of meridian inplane stress resultant, transverse shear, and slope, respectively. The strain-displacement measures and shell quantities are interpreted according to the shell type considered (i.e., shell of revolution or cylinder in plane strain).

5. FINITE ELEMENT SOLUTIONS FOR SHELLS OF REVOLUTION

The variational problem defined by Eqs. (4.1) and (4.2) along with Eqs. (2.2), (2.8) and (2.9) defines the behavior of shells of revolution. If the shell is used as a motor case in which a grain is contained and the grain analysis is carried out by a finite element analysis with polygonal elements, then the shell must also be composed of straight meridianal segments (i.e., conical frustrums); thus, in the development to follow $R \to \infty$ and φ remains constant within each element. Accordingly, the variational problem may be rewritten as

$$V = \int\int_B \left(\frac{1}{2} S_i B_{ij} S_j - S_i Q_i\right) dv - \int_C S_i F_i dc \qquad (5.1)$$

where

$$S_i = \left(u; w; M_{11j} u, s; w; s; M_{11}; s; \frac{u}{r}; \frac{w}{r}; \frac{1}{r}\frac{\partial u}{\partial s}; \frac{1}{r^2}\frac{\partial w}{\partial s}\right)$$

$$Q_i = (q_i; q_n; 0; \ldots\ldots 0) \qquad (5.2)$$

$$F_i = (N_{11}{}^a; Q_1{}^a; \theta^a; 0\ldots\ldots; 0)$$

and B_{ij} is a symmetric array whose non zero components are

$$B_{3,3} = -1/D_{11}$$

$$B_{3,9} = \sin\varphi$$

$$B_{3,10} = (1 - D_{12}/D_{11}) \cos\varphi$$

$$B_{4,4} = \bar{A}_{11}$$

$$B_{4,7} = A_{12} \cos \varphi$$

$$B_{4,8} = A_{12} \sin \varphi$$

$$B_{5,6} = 1.0$$

$$B_{7,7} = A_{22} \cos^2 \varphi$$

$$B_{7,8} = A_{22} \sin \varphi \cos \varphi$$

$$B_{7,11} = D_{22} \cos^2 \varphi$$

$$B_{8,8} = \bar{A}_{22} \sin^2 \varphi$$

$$B_{8,11} = \sin \varphi \cos \varphi \, D_{22}$$

$$B_{9,10} = \sin \varphi \cos \varphi \, D_{12}$$

$$B_{10,10} = \bar{D}_{22} \cos^2 \varphi$$

The finite element solution is achieved by subdividing the shell into subregions (elements) and approximating the primary dependent variable behavior in each element.

Consider

$$V_m' = \int_{s_m} \left(\frac{1}{2} S_i B_{ij} S_j - S_i Q_i \right) dv \tag{5.4}$$

then

$$V = \sum_{m=1}^{N-1} V_m' - \int_c S_i F_i dc \tag{5.5}$$

In element m take

$$u = \alpha_1 + \alpha_2 s$$

$$w = \alpha_3 + \alpha_4 s \qquad (5.6)$$

$$M = \alpha_5 + \alpha_6 s$$

and let

$$S_i = G_{ijk} \; \alpha_k X_j \qquad (5.7)$$

where

$$\alpha_i = (\alpha_1, \; \alpha_2, \ldots\ldots, \alpha_6) \qquad (5.8)$$

and

$$X_j = (1; \; s; \; \frac{1}{r}; \; \frac{s}{r}; \; \frac{1}{r^2})$$

The non zero components of G_{ijk} are

$$G_{111}=G_{122}=G_{213}=G_{224}=G_{315}=G_{326}=G_{512}=G_{614}=G_{716}$$

$$= G_{831}=G_{842}=G_{933}=G_{944}=G_{10,3,4}=G_{11,5,4} = 1 \qquad (5.9)$$

The functional may now be expressed as

$$V = \sum_{M=1}^{N-1} \left\{ \int_{s_m}^{s_{m+1}} \frac{1}{2} G_{ijk} \; \alpha_k \; B_{i\ell} G_{\ell mn} \alpha_n X_m \; dv \right.$$

$$\left. - \int_{s_m}^{s_{m+1}} G_{ijk} \; \alpha_k \; X_j \; Q_{id} \right\} - \int_c G_{ijk} \; \alpha_k \; X_j \; F_i \; dc \qquad (5.10)$$

Since all quantities are constants except X_j, the volume integrals may be evaluated leading to

$$V = \sum_{m=1}^{N-1} \left(\frac{1}{2} \alpha_i^{(m)} k_{ij}^{(m)} \alpha_j^{(m)} - \alpha_i^{(m)} f_i^{(m)} \right) - \alpha_1^{(c)} \bar{t}_i \tag{5.11}$$

where

$$H_{ij}^{(m)} = \int_{s_m}^{s_{m+1}} X_i X_j \, dv$$

$$k_{kn}^{(m)} = G_{ijk} H_{jm}^{(m)} B_{ij}^{(m)} G_{\ell mn} \tag{5.12}$$

and

$$f_k^{(m)} = G_{ijk} H_{ij}^{(m)} Q_i$$

$$\bar{t}_k = G_{ijk} X_j F_i$$

The α_i are expressed in terms of nodal displacements and moments through

$$\alpha_i^{(m)} = L_{ij}^{(m)} \phi_j \qquad i = 1,6 \qquad j = 1,3N \tag{5.13}$$

where

$$\phi_j = (u_1, w_1, M_1, u_2, w_2, M_2, \ldots\ldots u_N, w_N, M_N)$$

and

$$L_{i,j}^{(m)} = \frac{s_{m+1}}{s_{m+1} - s_m} \qquad \chi = 1,3$$

$$L_{i,j+3}^{(m)} = \frac{-s_m}{s_{m+1} - s_m} \quad ; \quad i = 2\chi - 1 \tag{5.14}$$

$$L_{i+1,j}^{(m)} = \frac{-1}{s_{m+1} - s_m} \qquad j = 3m-3+k$$

$$L_{i+1,j+3}^{(m)} = \frac{1}{s_{m+1} - s_m}$$

all other $L_{ij}^{(m)}$ are zero.

The functional takes the final form

$$V = \frac{1}{2} \phi_i K_{ij} \phi_j - \phi_i N_i - \phi_i \bar{T}_i \tag{5.15}$$

where

$$K_{ij} = \sum_{m=1}^{N-1} L_{ni}^{(m)} k_{np}^{(m)} L_{pj}^{(m)}$$

$$N_i = \sum_{m=1}^{N-1} L_{ni}^{(m)} f_n^{(m)} \tag{5.16}$$

and

$$\bar{T}_i = L_{ni}^{(c)} \bar{t}_n$$

The application of the direct method of the calculus of variations (Ritz Method) leads to the set of 3N equations to be solved for ϕ_i, the nodal values of the primary dependent variables; accordingly

$$\frac{\partial V}{\partial \phi_i} = 0 \qquad i = 1,\ldots,3N \tag{5.17}$$

which yields upon noting that $K_{ij} = K_{ji}$,

$$K_{ij} \phi_j - N_i - \bar{T}_i = 0 \qquad \begin{array}{l} i = 1,\ldots,3N \\ j = 1,\ldots,3N \end{array} \tag{5.18}$$

The force resultants are computed from Eq. (2.3) where displacements are obtained in terms of Equations (5.6), (5.13), (5.14) and the solution to (5.18).

6. FINITE ELEMENT SOLUTION FOR CYLINDRICAL SHELL

The steps carried out in the previous section are repeated herein for the cylindrical shell in a state of plane strain. Accordingly, the functional of the variational problem is given by

$$V = \iint_B \left(\frac{1}{2} S_i B_{ij} S_j - S_i Q_i\right) dv - \int_C S_i F_i \, dC$$

where

$$S_i = (q_i ; q_n ; 0 ; \ldots\ldots\ldots 0)$$

$$F_i = (N_{11}^a , Q_1^a , \theta_1^a , 0 , \ldots\ldots 0)$$

and B_{ij} is the symmetric array whose non-zero components are

$$B_{16} = c$$

$$B_{22} = c^2 A_{11}$$

$$B_{23} = c^2$$

$$B_{24} = cA_{11}$$

$$B_{33} = - 1/D_{11}$$

$$B_{34} = c$$

$$B_{44} = A_{11}$$

$$B_{56} = -1$$

The finite element solution is given by Eqs. (5.4) to (5.8) where

$$X_j = (1, s)$$

and the non-zero G_{ijk} are

$$G_{111}=G_{122}=G_{211}=G_{222}=G_{311}=G_{322}=G_{412}=G_{514}=G_{616} = 1.0$$

The remainder of the finite element description is given in Eqs. (5.10) to (5.18). The force resultants are computed from Eq. (3.3).

7. NUMERICAL EXAMPLE

In order to illustrate the results obtained by the development given herein, a cylindrical shell as shown in Fig. 2 and subjected to a constant edge shear is considered. This shell has been considered previously in [1] and [2] and hence a comparison can be made between the two different developments.

The same mesh characteristics as used in [2] are considered herein. These are shown in Fig. 2. For the present development the results for the transverse displacement and meridian moment are shown in Figs. 3 and 4. For comparison with [2], the transverse tip displacement and maximum meridian moment are tabulated vs. the mesh spacing over the first one inch of meridian distance from the loaded end. These results are shown in Table 1. As one might expect from the basis developed herein, the meridian moment is improved with a corresponding sacrifice in displacement accuracy. If, however, the true behavior is accurately protrayed everywhere along the structure, then both developments appear to give comparable results. Whether this means that incompatabilities between shell and elasticity elements are not important cannot be answered from the results of this example. The constraints imposed by a nearly incompressible elastic matrix may be extremely critical to the shell analysis.

8. CONCLUSION AND CLOSING REMARKS

A finite element shell analysis technique suitable for solid propellant rocket motor case analysis has been developed herein for shells of revolution subjected to axisymmetric loadings and cylinders in plane strain. The principal advantage of the present development over previous efforts is that it has a compatible displacement field with conventionally used elasticity finite elements. Thus, the effect of geometric incompatibilities is eliminated with the present development. Numerical efforts conducted to date for shell behavior show the present development to give comparable results with previous efforts provided a suitable mesh is selected. To date the effects of the shell-continua incompatibility in the previous works have not been assessed; however, the fact that no incompatability exists with the new development allows investigation of more salient topics in solid rocket integrity with the understanding that compatibility requirements are fulfilled between the motor case and the grain.

REFERENCES

1. Grafton, P.E. and Strome, P.R., "Analysis of Axisymmetrical Shells by the Direct Stiffness Method," AIAA J., Vol. 1, No. 10, October 1963, pp. 2342.

2. Percy, J.H., Pran, T.H.H., Klein, S., and Navaratna, D.R., "Application of Matrix Displacement Method to Linear Elastic Analysis of Shells of Revolution," AIAA J., Vol. 3., No. 11, November 1965, pp. 2138.

3. Popov, E.P., Penzien, J., and Lu, Z.A., "Finite Element Solution for Axisymmetrical Shells," J. Engr. Mech., ASCE, Vol. 90, EM 5, October 1964, pp. 119.

4. Novozhilov, V.V., The Theory of Thin Shells, P. Noordhoff Ltd., Gronmgen, 1959, (Translated by P.G. Lowe).



FIG. 1 FINITE ELEMENT IDEALIZATION OF SHELL



FIG. 2 ISOTROPIC CYLINDER WITH EDGE LOAD



FIG. 3 R - DISPLACEMENT



FIG. 4 MERIDIAN MOMENT (SHERM)

TABLE I

Case	Mesh (inches)	Tip Displacement G/S	Tip Displacement SHERM	Maximum Moment GS/10^2	Maximum Moment SHERM
4	0.5	2.62	1.57	.02	1.39
3	0.2	2.81	2.54	4.88	5.30
2	0.1	2.86	2.79	5.29	5.35
1	0.05	2.87	2.85	5.56	5.59
Exact		2.874	2.874	5.606	5.606

APPENDIX A

Computer Program Listing

```
$IBFTC HASE    DECK,LIST,REF
C      THIS DEVELOPEMENT  IS BASED ON A
C      FINITE ELEMENT SHELL ANALYSIS
C
C      DEPENDENT VARIABLES ARE U, W, M
C      WHERE EXPANSIONS ARE
C
C      U=A0+A1*S
C      W=A2+A3*S
C      M=A4+A5*S
C
C      S  IS  MERIDIAN  COORDINATE
C      U  IS  MERIDIAN  DISPLACEMENT
C      W  IS  TRANSVERSE  DISPLACEMENT
C      M  IS  MERIDIAN  MOMENT
       COMMON/COM1/XK(480,21),F(480),MBAND,NK
       COMMON/COM2/NP(3,160),R(160),Z(160),T(160),BC(160),FI(3,160),
      1 TE(160),FF(6),P,NUMEL
       COMMON/COM3/E(6,12),        TITLE(12)
C
   10  READ(5,1000) TITLE,NUMNP,NUMEL,NUMMAT
       WRITE(6,2000) TITLE,NUMNP,NUMEL,NUMMAT
       READ(5,1001) (M,(E(I,M), I=1,5),M=1,NUMMAT)
       DO 50 M=1,NUMMAT
       E(6,M)=E(5,M)
       E(5,M)=E(4,M)
   50  E(4,M)=E(2,M)*E(3,M)/E(1,M)
       WRITE(6,2001)
       WRITE(6,2002) (M,(E(I,M),I=1,6),M=1,NUMMAT)
       WRITE(6,2003)
C********************************
C      READ NODAL INFORMATION   *
C********************************
       READ(5,1002) (M,BC(M),R(M),Z(M),T(M),TE(M),(FI(I,M),I=1,3),
      1 M=1,NUMNP)
       WRITE(6,2004) (M,BC(M),R(M),Z(M),T(M),TE(M),(FI(I,M),I=1,3),
      1 M=1,NUMNP)
C*****************************
C      INITIALIZE MATRICES   *
C*****************************
       NK=3*NUMNP
       DO 60 II=1,NK
       F(II)=0.0
       DO 60 JJ=1,21
   60  XK(II,JJ)=0.0
       MBAND=0
C*********************************************************
C      READ ELEMENT CARDS AND FORM STIFFNESS MATRIX     *
C*********************************************************
       WRITE(6,2005)
   70  READ(5,1003) NEL,I,J,MAT,P
       WRITE(6,2008) NEL,I,J,MAT,P
       NP(1,NEL)=I
       NP(2,NEL)=J
       NP(3,NEL)=MAT
       N1=NEL
       CALL SHERM(NEL,I,J,MAT)
```

```
      MK=IABS(3*(I-J))+3
      IF(MBAND.LT.MK) MBAND=MK
      IF (N1.LT.NUMEL) GO TO 70
C*****    ADD CONCENTRATED FORCES TO LOAD VECTOR
C****************************************
C*****    MODIFY FOR BOUNDARY CONDITIONS
C****************************************
      DO 100 N=1, NUMNP
      N1=3*N
      IF(BC(N).LT.0.0) GO TO 130
      NN=BC(N)+1.0
      GO TO (105,110,115,120,125) , NN
C*****    FREE SUPPORT/SPECIFIED MOMENT
  110 IQ=3
      U=FI(IQ,N)
      CALL MODIFY (N,IQ,U)
      F(N1-2)=F(N1-2)+FI(1,N)
      F(N1-1)=F(N1-1)+FI(2,N)
      GO TO 100
C*****    ROLLER SUPPORT
  115 DO 200 IQ=2, 3
      U=FI(IQ,N)
      CALL MODIFY (N,IQ,U)
  200 CONTINUE
      F(N1-2)=F(N1-2)+FI(1,N)
      GO TO 100
C*****    PIN SUPPORT
  120 DO 205 IQ=1, 3
      U=FI(IQ,N)
      CALL MODIFY (N,IQ,U)
  205 CONTINUE
      GO TO 100
C*****    CLAMPED SUPPORT
  125 DO 210 IQ=1, 2
      U=FI(IQ,N)
      CALL MODIFY (N,IQ,U)
  210 CONTINUE
      F(N1)=F(N1)+FI(3,N)
      GO TO 100
  130 BVAL=BC(N)
      FVAL=FI(2,N)
      CALL NODIFY(N,FVAL,BVAL)
      GO TO 100
  105 F(N1-2)=F(N1-2)+FI(1,N)
      F(N1-1)=F(N1-1)+FI(2,N)
      F(N1)=F(N1)+FI(3,N)
  100 CONTINUE
      CALL BANSOL(NK,MBAND)
C*********************************
C     PRINT OUTPUT OF ANSWERS     *
C*********************************
      WRITE(6,2006) TITLE
      WRITE(6,2007) (N,R(N),Z(N),F(3*N-2),F(3*N-1),F(3*N),N=1,NUMNP)
      WRITE(6,2009)
      CALL STRESS
      GO TO 10
 1000 FORMAT (12A6/3I5)
 1001 FORMAT (I5,5E10.0)
```

```
1002 FORMAT (I5,F5.0,4F10.0,3E10.0)
1003 FORMAT(4I5,F10.0)
2000 FORMAT(1H1,12A6/
    1 25HONUMBER OF NODES---------=  I5/
    2 25HONUMBER OF ELEMENTS------=  I5/
    3 25HONUMBER DIFF. MATERIALS-=  I5//)
2001 FORMAT (20HOMATERIAL PROPERTIES//10H  MATERIAL ,20X,
    1 8HMERIDIAN,28X,15HCIRCUMFERENTIAL//
    2 23X,7HMODULUS,7X,13HPOISSON RATIO,13X,7HMODULUS,7X,
    3 13HPOISSON RATIO,13X,7HALPHA11,13X,7HALPHA22//)
2002 FORMAT (I10,6E20.5)
2003 FORMAT (5H1NODE,6X,4HB.C.,4X,8HR-COORD.,4X,8HZ-COORD.,3X,
    1 9HTHICKNESS,1X,11HTEMPERATURE,3X,17HR FORCE OR DISPL.,3X,
    2 17HZ FORCE OR DISPL.,3X,17HM FORCE OR DISPL.//)
2004 FORMAT (I5,0P1F10.2,0P4F12.4,1P3E20.5)
2005 FORMAT (10H1  ELEMENT,4X,6HI-NODE,4X,6HJ-NODE,2X,8HMATERIAL,2X,
    18HPRESSURE//)
2006 FORMAT (1H1,12A6//5H NODE,9X,1HR,9X,1HZ,6X,14HR-DISPLACEMENT,6X,
    1 14HZ-DISPLACEMENT,5X,15HMERIDIAN MOMENT//)
2007 FORMAT(I5,2F10.2,3E20.5)
2008 FORMAT(I7,3I10,F10.0)
2009 FORMAT(31H1ELEMENT      I/J  COORDINATES  30X33HS T R E S S   R E S
    1 U L T A N T S/19X9HR       Z10X13HLONG. RESULT.7X13HCIRC. RESULT.
    17X13HSHEAR RESULT.8X12HLONG. MOMENT8X12HCIRC. MOMENT////)
     END
```

```
$IBFTC SHEL    DECK,LIST,REF
      SUBROUTINE SHERM(NN,I,J,MAT)
      COMMON/COM1/XK(480,21),F(480),MBAND,NK
      COMMON/COM2/NP(3,160),R(160),Z(160),T(160),BC(160),FI(3,160),
     1 TE(160),FF(6),P,NUMEL
      COMMON/COM3/E(6,12),         TITLE(12)
      DIMENSION S(6,6), LM(2)
      REAL L
      A=Z(I)-Z(J)
      B=R(J)-R(I)
      L=SQRT(A**2+B**2)
      TA=0.5*(T(I)+T(J))
      THERM=0.5*(TE(I)+TE(J))
      COMM=1.-E(2,MAT)*E(4,MAT)
      A11=E(1,MAT)*TA/COMM
      A12=E(4,MAT)*A11
      A22=E(3,MAT)*TA/COMM
      D11=A11*TA**2/12.
      D12=A12*TA**2/12.
      D22=A22*TA**2/12.
      D22B=D22-D12**2/D11
C
C     INITIALIZATION
C
      DO 1 II=1,6
      DO 1 JJ=1,6
    1 S(II,JJ)=0.0
C*********************************************************************
C     EVALUATION OF INTEGRALS
C*********************************************************************
      XI1=L*(R(J)+R(I))/2.0
      XI2=L
      XI3=L/2.
      XI7=(R(I)+2.*R(J))*L/6.0
      XI8=(R(I)+3.*R(J))*L/12.0
      IF (B.EQ.0.0) GO TO 100
      IF(R(I).EQ.0..OR.R(J).EQ.0.)GO TO 90
      XI4=L*ALOG(R(J)/R(I))/B
      GO TO 92
   90 IF(R(I).EQ.0.)GO TO 91
      XI4=L*ALOG(.000001)/B
      GO TO 92
   91 XI4=L*ALOG(1000000.)/B
   92 XI5=(XI2-R(I)*XI4)/B
      XI6=(XI1-2.*R(I)*XI2+R(I)**2*XI4)/B**2
      GO TO 110
  100 XI4=XI2/R(I)
      XI5=XI3/R(I)
      XI6=L/(3.*R(I))
  110 CONTINUE
C
C     ELEMENT STIFFNESS MATRIX
C
      TEMP=XI4*(B/L)**2*D22B
      S11=A11*XI1-2.*A12*B*(XI2-XI3)+A22*B**2*(XI4-2.*XI5+XI6)
      S12=-A11*XI1-A12*B*(2.*XI3-XI2)+A22*B**2*(XI5-XI6)
      S13=-A12*A*(XI2-XI3)+A22*A*B*(XI4-2.*XI5+XI6)
```

```
      S14=-A12*A*XI3+A22*A*B*(XI5-XI6)
      S15=0.
      S16=0.
      S22=A11*XI1+2.*A12*B*XI3+A22*B**2*XI6
      S23=A12*A*(XI2-XI3)+A22*A*B*(XI5-XI6)
      S24=A12*A*XI3+A22*A*B*XI6
      S25=0.
      S26=0.
      S33=A22*A**2*(XI4-2.*XI5+XI6)+TEMP
      S34=A22*A**2*(XI5-XI6)-TEMP
      S35=XI1-(XI2-XI3)*B*(1.-D12/D11)
      S36=-XI1-XI3*B*(1.-D12/D11)
      S44=A22*A**2*XI6+TEMP
      S45=-S35
      S46=-S36
      S55=-(XI1-2.*XI7+XI8)*L*L/D11
      S56=-(XI7-XI8)*L*L/D11
      S66=-XI8*L*L/D11
C
C*****    ADD TEMPERATURE AND PRESSURE EFFECTS TO LOAD VECTOR
C
      FF(1)=THERM*(-XI1/L*(A11*E(5,MAT)+A12*E(6,MAT))+B/2.*(A12*E(5,MAT)
     1+A22*E(6,MAT)))
      FF(2)=THERM*A/2.*(A12*E(5,MAT)+A22*E(6,MAT))+P*L*(R(I)/2.+B/6.)
      FF(4)=THERM*(XI1/L*(A11*E(5,MAT)+A12*E(6,MAT))+B/2.*(A12*E(5,MAT)+
     1 A22*E(6,MAT)))
      FF(5)=THERM*A/2.*(A12*E(5,MAT)+A22*E(6,MAT))+P*L*(R(I)/2.+B/3.)
C
C     TRANSFORM ELEMENT STIFFNESS TO GLOBAL COORDINATE SYSTEM
C
      A=A/L
      B=B/L
      S(1,1)=S11*B**2+2.*S13*A*B+S33*A**2
      S(1,2)=-S11*A*B+S13*B**2-S13*A**2+S33*A*B
      S(1,3)=S35*A
      S(1,4)=S12*B**2+S14*A*B+S23*A*B+S34*A**2
      S(1,5)=-S12*A*B+S14*B**2-S23*A**2+S34*A*B
      S(1,6)=S36*A
      S(2,2)=S11*A**2-2.*S13*A*B+S33*B**2
      S(2,3)=S35*B
      S(2,4)=-S12*A*B-S14*A**2+S23*B**2+S34*A*B
      S(2,5)=S12*A**2-S14*A*B-S23*A*B+S34*B**2
      S(2,6)=S36*B
      S(3,3)=S55
      S(3,4)=S45*A
      S(3,5)=S45*B
      S(3,6)=S56
      S(4,4)=S22*B**2+S24*A*B*2.+S44*A**2
      S(4,5)=-S22*A*B+S24*B**2-S24*A**2+S44*A*B
      S(4,6)=S46*A
      S(5,5)=S22*A**2-2.*S24*A*B+S44*B**2
      S(5,6)=S46*B
      S(6,6)=S66
      DO 3 II=1,5
      IP1=II+1
      DO 3 JJ=IP1,6
    3 S(JJ,II)=S(II,JJ)
      DO 5 II=1,6
```

```
      DO 5 JJ=1,6
    5 S(II,JJ)=S(II,JJ)/L**2
C*****     TRANSFORM LOAD TO GLOBAL COORDS)
      T1=FF(1)
      T2=FF(4)
      FF(1)=FF(1)*B+FF(2)*A
      FF(2)=FF(2)*B-T1*A
      FF(4)=FF(4)*B+FF(5)*A
      FF(5)=FF(5)*B-T2*A
C***********************************************************
C     ADD ELEMENT STIFFNESS TO TOTAL STIFFNESS            *
C***********************************************************
      LM(1)=3*I-3
      LM(2)=3*J-3
      DO 10 IN=1,2
      DO 10 KN=1,3
      II=LM(IN)+KN
      KK=3*IN-3+KN
      F(II)=F(II)+FF(KK)
      DO 10 JN=1,2
      DO 10 LN=1,3
      JJ=LM(JN)+LN-II+1
      LL=3*JN-3+LN
      IF(JJ)10,10,9
    9 XK(II,JJ)=XK(II,JJ)+S(KK,LL)
   10 CONTINUE
      RETURN
      END
$IBFTC MODI     DECK,LIST,REF
C*****     MODIFICATION FOR BOUNDARY CONDITIONS
      SUBROUTINE MODIFY (N,I,X)
      COMMON/COM1/ A(480,21), B(480), MM, NN
      NI=3*N+I-3
      DO 250 M=2, MM
      K=NI-M+1
      IF (K.LE.0) GO TO 235
      B(K)=B(K)-A(K,M)*X
      A(K,M)=0.0
  235 K=NI+M-1
      IF (NN.LT.K) GO TO 250
      B(K)=B(K)-A(NI,M)*X
      A(NI,M)=0.0
  250 CONTINUE
      A(NI,1)=1.0
      B(NI)=X
      RETURN
      END
$IBFTC NODI     DECK,LIST,REF
      SUBROUTINE NODIFY(N,X,Y)
      COMMON/COM1/A(480,21),B(480),MM,NN
      DX=COS(Y)
      DY=SIN(Y)
      NI=3*N-3
      DO 250 M=2,MM
      K=NI-M+2
      IF(K.LE.0) GO TO 235
      B(K)=B(K)-X*(A(K,M+1)*DX-A(K,M)*DY)
      A(K,M)=A(K,M)*DX+A(K,M+1)*DY
```

```
      A(K,M+1)=0.0
  235 K=NI+M+1
      IF(NN.LT.K) GO TO 250
      B(K)=B(K)+X*(A(NI+1,M+1)*DY-A(NI+2,M)*DX)
      A(NI+1,M+1)=A(NI+1,M+1)*DX+A(NI+2,M)*DY
      A(NI+2,M)=0.0
  250 CONTINUE
      A(NI+1,1)=A(NI+1,1)*DX*DX+A(NI+2,1)*DY*DY+2.0*DX*DY*A(NI+1,2)
      A(NI+2,1)=1.0
      B(NI+2)=X
      RETURN
      END
```

```
$IBFTC BANS     DECK,LIST,REF
C*****************************************************************************
      SUBROUTINE BANSOL(NN,MM)
C*****************************************************************************
      COMMON/COM1/ A(480,21),F(480),MM,NN
      CALL TIME(Z,NNN)
      DO 280 N=1,NN
      DO 260 L=2,MM
      C=A(N,L)/A(N,1)
      I=N+L-1
      IF (NN.LT.I) GO TO 260
      J=0
      DO 250 K=L,MM
      J=J+1
  250 A(I,J)=A(I,J)-C*A(N,K)
      F(I)=F(I)-C*F(N)
  260 A(N,L)=C
  280 F(N)=F(N)/A(N,1)
C     BACKSUBSTITUTION
      N=NN
  300 N=N-1
      IF (N.LE.0) GO TO 500
  350 DO 400 K=2,MM
      L=N+K-1
      IF (NN.LT.L) GO TO 400
      F(N)=F(N)-A(N,K)*F(L)
  400 CONTINUE
      GO TO 300
  500 CALL TIME(Z,MMM)
      MMM=MMM-NNN
      PRINT 100, MMM
      RETURN
  100 FORMAT (16H0TIME IN BANSOL=,I10,13H MILLISECONDS/)
      END
$IBFTC STRS     DECK,LIST,REF
      SUBROUTINE STRESS
      COMMON/COM1/XK(480,21),F(480),MBAND,NK
      COMMON/COM2/NP(3,160),R(160),Z(160),T(160),BC(160),FI(3,160),
     1 TE(160),FF(6),P,NUMEL
      COMMON/COM3/E(6,12),        TITLE(12)
      REAL L
      DO 100 N=1,NUMEL
      I=NP(1,N)
      J=NP(2,N)
      MAT=NP(3,N)
      UI=F(3*I-2)
      WI=F(3*I-1)
      UJ=F(3*J-2)
      WJ=F(3*J-1)
      A=Z(I)-Z(J)
      B=R(J)-R(I)
      L=SQRT(A**2+B**2)
      TA=0.5*(T(I)+T(J))
      COMM=1.-E(2,MAT)*E(4,MAT)
      A11=E(1,MAT)*TA/COMM
      A12=E(4,MAT)*A11
      A22=E(3,MAT)*TA/COMM
```

```
      D11=A11*TA**2/12.
      D12=A12*TA**2/12.
      D22=A22*TA**2/12.
      D22B=D22-D12**2/D11
      X1I=UI*(-A11*B/L**2+A12/R(I))+UJ*A11*B/L**2+A11*A/L**2*(WI-WJ)
      X1J=-UI*A11*B/L**2+UJ*(A11*B/L**2+A12/R(J))+A11*A/L**2*(WI-WJ)
      X2I=UI*(-A12*B/L**2+A22/R(I))+UJ*A12*B/L**2+A12*A/L**2*(WI-WJ)
      X2J=-UI*A12*B/L**2+UJ*(A12*B/L**2+A22/R(J))+A12*A/L**2*(WI-WJ)
      Y2I=D12*F(3*I)/D11-D22B*B*(A*(UJ-UI)+B*(WJ-WI))/(R(I)*L**3)
      Y2J=D12*F(3*J)/D11-D22B*B*(A*(UJ-UI)+B*(WJ-WI))/(R(J)*L**3)
      QI=B*F(3*I)*(1.-D12/D11)/(R(I)*L)-D22B*B**2*(A*(UJ-UI)+B*(WJ-WI))/
     1(R(I)**2*L**4)+(F(3*J)-F(3*I))/L
      QJ=B*F(3*J)*(1.-D12/D11)/(R(J)*L)-D22B*B**2*(A*(UJ-UI)+B*(WJ-WI))/
     1(R(J)**2*L**4)+(F(3*J)-F(3*I))/L
  100 PRINT 110,N,I,R(I),Z(I),X1I,X2I,QI,F(3*I),Y2I,J,R(J),Z(J),X1J,X2J
     1,QJ,F(3*J),Y2J
  110 FORMAT(1H0I5,I9,0P2F8.3,1P5E20.7/I15,0P2F8.3,1P5E20.7//)
      RETURN
      END
```

APPENDIX B - COMPUTER USER'S MANUAL

A computer code for the class of problems considered in Sections 2 and 5 (shells of revolution subjected to axisymmetric loadings) is included in this appendix. The computer code is written in Fortran IV and may be utilized with the IBM 7094 or equivalent computer. The computations are conducted in-core and no auxiliary storage is required. The code has a capacity for 160 nodes on the shell.

The following serves as input information for the finite element description of the shell. After dividing the shell into an appropriate mesh, the nodes are numbered in sequence along the meridian (or at least so that the difference between numbers of adjacent nodes differs by 6 or less, which allows for analysis of toroidal shells). The following data cards are then punched as input information:

1) Title Card (12A6)

This is printed as output to define the problem considered.

2) Control Card (3I5)

Col.	Information	Format
1-5	Number of Nodes (not to exceed 160)	I
6-10	Number of Elements (not to exceed 160)	I
11-15	Number Different Materials (less than 13)	I

3) Material Cards (I5, 5F10.0)

One for each different material

Col.	Information	Format
1-5	Material Number less than 13	I
6-15	Modulus along Meridian	F
16-25	Poisson ratio along Hoop	F
26-35	Modulus along Hoop	F
36-45	Thermal Expansion - Meridian	F
46-55	Thermal Expansion - Hoop	F

The second poisson ratio is computed internally in the program.

4) Nodal Cards (I5, F5.0, 7F10.0)

One card per node

Col.	Information	Format
1-5	Node Number	I
6-10	Boundary Code	F
11-20	R ordinate	F
21-30	Z ordinate	F
31-40	Thickness	F
41-50	Temperature	F
51-60	R-Force or Displacement	F
61-70	Z-Force or Displacement	F
71-80	Moment or Rotation	F

The information in Cols. 51-80 is interpreted according to the number punched in Col. 10.

No. in Col. 10	Col. 51-60	Col. 61-70	Col. 71-80
0	R-Force	Z-Force	Rotation
1	R-Force	Z-Force	Moment
2	R-Force	Z-Displ	Moment
3	R-Displ	Z-Displ	Moment
4	R-Displ	Z-Displ	Rotation

If the number in Cols. 6-10 is negative then the support is taken as a roller support with the axes makeup an angle with the R-axis, Col. 51-60 is the force along the incline Col. 61-70 the normal displacement to the incline Col. 71-80 the specified moment.

5) Element Cards (4I, F10.0)

One per element

Col.	Information	Format
1-5	Element Number	I
6-10	I-Node	I
11-15	J-Node	I
16-20	Material Number	I
21-30	Pressure Load on Shell	F

The following is printed as output of the program:

1. Input Information
2. Shell Displacements and Meridian Moment
3. The Shell Force Resultants

Unclassified
Security Classification

DOCUMENT CONTROL DATA - R&D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)*

University of California
Berkeley, California

2a. REPORT SECURITY CLASSIFICATION
Unclassified

2b. GROUP

3. REPORT TITLE

Finite Element Analysis for Solid Rocket Motor Cases

4. DESCRIPTIVE NOTES *(Type of report and inclusive dates)*

5. AUTHOR(S) *(Last name, first name, initial)*

Herrmann, Leonard R., Taylor, Robert L., Green, David R.

6. REPORT DATE
March 1967

7a. TOTAL NO. OF PAGES

7b. NO. OF REFS
4

8a. CONTRACT OR GRANT NO.
Subcontract No. B-87010-US

b. XXXXXXXXX
Prime Contract N123(60530)-55641A

c.

d.

9a. ORIGINATOR'S REPORT NUMBER(S)

SEL 67-4

9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*

10. AVAILABILITY/LIMITATION NOTICES

Unlimited distribution

11. SUPPLEMENTARY NOTES

12. SPONSORING MILITARY ACTIVITY

13. ABSTRACT

A finite element development for shells of revolution and cylinders in plane strain is presented. The element is based upon a variational theorem which allows linear displacement expansions in each element, thus, an element is developed which is fully compatible with commonly employed continuum elements. A numerical example is presented along with a computer program listing for shells of revolution.